

Лабораторная работа № 5
Определение критических сил и форм волнообразования
при потере устойчивости пологой арки
РЕШЕНИЕ ЗАДАЧ УСТОЙЧИВОСТИ В ANSYS

КЛЮЧЕВЫЕ СЛОВА

1. Анализ потери устойчивости
2. Нелинейный анализ

ОПИСАНИЕ ЗАДАЧИ

Необходимо исследовать устойчивость конструкции вместе с приложенными нагрузками. Провести анализ задачи в ANSYS по следующим методикам:

- 1) по методу определения бифуркационных значений нагрузок из линеаризованных уравнений устойчивости (Buckling Eigenvalue Analysis)
- 2) по методу определения траектории равновесных состояний «нагрузка-перемещение» из нелинейных соотношений (Nonlinear Buckling Analysis)

Условие задачи.

Рассмотрим задачу о прохлопывании пологой арки. Пологая арка с диаметром основания D и высотой купола H имеет прямоугольное поперечное сечение с высотой h и толщиной b . Арка нагружена сверху равномерно распределенной нагрузкой интенсивности q и жестко закреплена по основаниям. Данная арка, отнесенная к декартовой системе координат Oxy , показана на рис. 1. Примем следующие значения для входных данных задачи: $D=20$ см; $H=3.2$ см; $h=0.3$ см; $b=1$ см; $E=2.02 \cdot 10^6$ кг/см²; $\nu=0.3$.




Рис. 1.

Методические указания

Конечно-элементный комплекс ANSYS позволяет проводить моделирование разнообразных задач механики и физики. В частности, ANSYS предоставляет средства для численного решения задач устойчивости деформируемых систем.

В качестве учебных вычислительных заданий предлагаются расчеты на устойчивость различных простых задач: прохлопывание пологих арок, прямоугольных панелей, конических и сферических панелей в осесимметричных постановках и др. ANSYS предоставляет возможность для проведения двух видов конечно-элементного анализа задач устойчивости.

Первый вариант состоит в построении линейного приближения и последующего определения точек бифуркации линеаризованных уравнений устойчивости (Buckling Eigenvalue Analysis). Простейшим примером такого подхода является классическая задача Эйлера о равновесии упругого стержня, продольно сжатого силами F .

В ANSYS построение линейного приближения состоит в работе решателя SOLUTION при статическом типе анализа с включенной опцией «преднапряженного» состояния

```
/solu  
antype, static  
pstres, on
```

Далее задаются условия закрепления и силовые факторы, которые могут вызывать потерю устойчивости. Величина этих силовых факторов в выбранной системе единиц обычно полагается равной единице для полной модели, либо, при расчете симметричных n -тых частей моделей, значениям $1/n$. (В противном случае следует запомнить множитель силового фактора и полученное в дальнейшем значение критической силы разделить на этот множитель.) При выполнении команды SOLVE формируется геометрическая или инкрементальная матрица жесткости S , пропорциональная приложенным силовым факторам.

Затем после выхода из решателя SOLUTION организуется второй этап, состоящий в решении задачи на собственные значения:

$$\mathbf{K} \cdot \mathbf{U} + \lambda \mathbf{S} \cdot \mathbf{U} = 0, \quad (1)$$

где \mathbf{K} - обычная матрица жесткости, \mathbf{S} - геометрическая матрица жесткости, определенная ранее, \mathbf{U} - глобальный вектор узловых степеней свободы.

Уравнения (1) являются КЭ аппроксимациями линеаризованных уравнений устойчивости для определения точек бифуркаций. Собственные числа λ_k являются (приближенными) значениями критических сил или нагрузок, а соответствующие собственные векторы \mathbf{U}_k описывают формы потери устойчивости, называемые также формами волнообразования. В задачах устойчивости, как правило, представляют интерес только первая пара $\{\lambda_1, \mathbf{U}_1\}$, но в ANSYS легко можно найти и последующие значения. Пример реализации этапа решения задачи (1) состоит в выполнении следующих команд:

```
/solu  
antype, buckle  
bucopt, subsp, NMODE  
...
```

```
solve
finish
```

где *NMODE* - заказываемое пользователем количество k пар $\{\lambda_k, \mathbf{U}_k\}$.

Помимо метода итераций в подпространстве (subsp), для решения задачи (1) на собственные значения ANSYS допускает и использование метода reduce, но не особенно его рекомендует, так как тогда надо определять главные степени свободы (master DOF), а для получения собственных форм использовать еще и шаг mxprand.

Доступ к вычисленным собственным значениям λ_k и формам \mathbf{U}_k может быть организован в постпроцессоре /post1 аналогично обработке результатов расчетов на собственные частоты (см. далее пример Buckl_Eigenvalue.inp).

Второй вариант анализа задач устойчивости (Nonlinear Buckling Analysis) более точный, но и требует значительных затрат машинного времени. Этот анализ основывается на нелинейных соотношениях и состоит в определении траектории равновесных состояний, т.е. зависимости «нагрузка-перемещение». Нелинейные КЭ-соотношения здесь имеют вид:

$$\mathbf{K}(\mathbf{U}) \cdot \mathbf{U} = \mathbf{F}. \quad (2)$$

От других видов нелинейного анализа задачи устойчивости отличаются тем, что здесь по предыдущим значениям пар $\{\mathbf{U}_n, \mathbf{F}_n\}$ и (2) требуется определять последующие пары $\{\mathbf{U}_{n+1}, \mathbf{F}_{n+1}\}$ равновесных состояний, причем \mathbf{F}_n и \mathbf{F}_{n+1} считаются пропорциональными. Для решения задачи (2) в ANSYS имеется специальный метод — Arc-Length Method, в котором осуществляются явные сферические итерации по ортогональным к радиусам дуг направлениям. Основная группа команд решателя SOLUTION, задающих нелинейный анализ устойчивости, следующая:

```
/solu
antype,static
nlgeom,on      ! Включение учета больших перемещений
                ! (геометрической нелинейности)
arclen,on      ! Включение Arc-Length метода
nsub, NSUBST   ! Задание параметра NSUBST для
                ! определения радиуса в Arc-Length методе как
                ! макс. значение нагрузки / NSUBSP
outres,all,all ! Запись всех выходных данных
                ! на всех шагах решения в базу данных
. . . .       ! Задание условий закрепления
                ! и максимальных значений нагрузок
solve          ! Запуск на счет
finish
```

Более подробную информацию о командах ANSYS, используемых в нелинейном анализе устойчивости, а также об опциях этих команд можно найти в соответствующих разделах документации по ANSYS. Пример постпроцессорной обработки результатов, связанной с построением кривой «нагрузка-перемещение», приводится далее в тексте программы Buckl_Nonlinear.inp.

РЕШЕНИЕ ЗАДАЧИ С ИСПОЛЬЗОВАНИЕМ ANSYS

В силу симметрии задачи относительно оси Oy достаточно рассматривать только правую половину арки при $x \geq 0$. В качестве КЭ выберем плоский балочный элемент BEAM3. Степенями свободы для этого элемента являются осевые перемещения UX , UY , и угол поворота $ROTZ$. В качестве граничных условий задачи примем: $UX=UY=ROTZ=0$ при $x=D/2$, $y=0$ (жесткая заделка); $UX=ROTZ=0$ при $x=0$, $y=H$ (условия симметрии). Так как рассматривается половина модели, то к правой половине арки надо прикладывать нагрузку, в два раза меньшую, чем к полной модели, т.е. $q/2$.

Приводимая ниже программа Buckl_Eigenvalue.inp находит первые собственные значения критического давления q и соответствующие формы волнообразования, используя первый вариант анализа устойчивости, т.е. Eigenvalue Buckling Analysis. Программа предназначена для выполнения в пакетном режиме ANSYS и снабжена подробными комментариями. Изменяя некоторые команды, как указывается в комментариях, можно исследовать также задачу с сосредоточенной силой, действующей в верхней точке купола арки.

```
!*****
!
!       Файл Buckl_Eigenvalue.inp
!       Определение критических сил и форм волнообразования
!       при потере устойчивости пологой арки.
!       Eigenvalue Buckling Analysis
!       (С учетом симметрии задачи рассматривается половина арки)
!       Условия нагружения - распределенная нормальная нагрузка
!
!                               или сосредоточенная сила в верхней точке
! купола
!*****
f_r='Buckl_p'           ! имя файлов для вывода результатов
nform=1                ! Количество первых значений критических сил
nldiv=40               ! Параметр для разбиения линий
! Все входные данные задаются в системе СГС
hl=0.3                 ! Высота поперечного сечения арки
bl=1                   ! Толщина поперечного сечения арки
hg=3.2                 ! Высота купола арки
darc=20                ! Диаметр основания арки

rarc=hg/2+darc*darc/8/hg ! Радиус кривизны арки
pi=4*atan(1)          ! Число PI
alfar=acos((rarc-hg)/rarc) ! угол раствора половины арки в
радианах
alfa=alfar*180/pi      ! угол раствора половины арки в
градусах
larc=2*alfar*rarc      ! длина арки
/output,f_r,res        ! Сервисный вывод в файл <f_r>.res
*vwrite
(1X,' Определение критических сил и форм волнообразования')
*vwrite
(1X,' при потере устойчивости пологой арки.')
```

```

(1X,' Eigenvalue Buckling Analysis ')
*vwrite
(1X)
*vwrite,darc,hg,h1,bl
(1X,' DARC=', E12.5,' HG=', E12.5,' HL=', E12.5,' BL=',
E12.5)
*vwrite,rarc
(1X,' RARC=', E12.5)
*vwrite
(1X)
/output          ! Окончание сервисного вывода

/prep7
et,1,beam3          ! КЭ ВЕАМ3
mp,ex,1,2.02e6      ! Модуль Юнга EX=2.02e6
mp,nuxy,1,0.3       ! Коэффициент Пуассона NUXY=0.3
r,1,bl*h1,(bl*h1**3)/12,h1 ! R. const для поперечного сечения
балки:
                                ! площадь, момент инерции, толщина
csys,1              ! Переход в цилиндрическую систему координат
k,1,rarc,90         ! Точка 1 - вершина купола
k,2,rarc,90-alfa   ! Точка 2 - основание арки
n,1,rarc,90        ! Узел 1 будет далее узлом приложения
силы,
n,2,rarc,90-alfa   ! узел 2 - основанием арки

l,1,2,nldiv         ! Генерация формы половины арки
! (В цилиндрической системе коорд. линии между точками
! с одинаковыми радиусами есть дуги окружности)
lmesh,all          ! Генерация сетки КЭ на линии
nummrg,node        ! Узлы 1,2 включаются в КЭ модель
finish
! Этап определения геометрической матрицы жесткости S
/solu
antype,static      ! Статический тип анализа
pstres,on          ! Включение расчета "предварительных напряжений"
(S)
dk,1,ux,0,,,rotz   ! Условия симметрии в вершине купола арки
dk,2,ux,0,,,uy,rotz ! Жесткая заделка опорной точки 2 арки
! В случае действия сосредоточенной силы команду с sbeam
следует
! закомментировать, а команду с f - раскомментировать.
!f,1,fy,-0.5       ! Задание сжимающей силы в верхней точке (1)
купола
! (Значение FY=-0.5, т.к. рассматривается половина модели)
sbeam,all,,pres,0.5 ! Задание давления
! (pres=0.5, т.к. рассматривается половина модели)
outpr,,1
solve
finish
! Этап определения точек бифуркации
! линеаризированной задачи устойчивости
/solu

```

```

antype,buckle          ! Тип анализа - решение лин. задач
устойчивости
bucopt,subsp,nform    ! Subspace-метод решения задачи
! на собственные значения с поиском nform собственных значений
outres,all,all        ! Запись всех выходных данных
! на всех шагах решения в базу данных
/psc,all,1            ! Показ всех граничных условий
/psf,pres,norm,1     ! Показ распределенных нагрузок
eplot                 ! Графический вывод конечно-элементной модели
solve
finish
/post1                ! Постпроцессорная обработка результатов
! В цикле номерам по собственным значениям считываются
! результаты для каждого собственного значения.
! Величины критических сил и/или нагрузок заносятся в файл
вывода и
! рисуется форма потери устойчивости
*do,i,1,nform
  set,,i
  *get,fmode,mode,i,freq
! Для сосредоточенной силы следующие три команды
! следует раскомментировать, а последующие четыре -
закомментировать

!   /output,F_r,res,,append
!   *vwrite,i,fmode
!   (1X,F4.0,' Force=',E12.5)
  vforce=fmode*larc
  /output,F_r,res,,append
  *vwrite,i,fmode,vforce
  (1X,F4.0,' Pressure=',E12.5,' Total force=',E12.5)
  /output
  pldisp,1
*enddo

```

В результате выполнения приведенной программы в случае действия распределенной нагрузки было получено значение первой критической нагрузки $q_1 = 156.14 \text{ кГ/см}^2$. Соответствующая первая форма волнообразования показана на рис. 2а.

Та же задача для арки в программе Buckl_Nonlinear.inp решается с использованием нелинейного анализа устойчивости. Полученное в предыдущей программе значение критической нагрузки q_1 может служить основанием для выбора верхнего предела изменения q при построении кривой «нагрузка-прогиб» как $q_{\max} = 160 \text{ кГ/см}^2$. (Тогда для половины модели верхнее значение есть $q_{\max} / 2$.) Текст программы приводится ниже.

```

!*****
!           Файл Buckl_Nonlinear.inp
!   Прохлопывание пологой арки.
!   Нелинейный анализ устойчивости.
!   Строится кривая равновесных состояний

```

```

! "нагрузка-прогиб" для верхней точки купола арки
!*****
f_r='Buckl_n'          ! имя файлов для вывода результатов
!vforce=450 ! Значение максимальной силы (половинное значение)
vpres=80 ! Значение максимального давления (половинное значение)
nldiv=20             ! Параметр для разбиения линий
! Все входные данные задаются в системе СГС
hl=0.3              ! Высота поперечного сечения арки
bl=1                ! Толщина поперечного сечения арки
hg=3.2              ! Высота купола арки
darc=20             ! Диаметр основания арки

rarc=hg/2+darc*darc/8/hg ! Радиус кривизны арки
pi=4*atan(1) ! Число PI
alfa=acos((rarc-hg)/rarc)*180/pi ! alfa - угол раствора половины
арки

/output,f_r,res      ! Сервисный вывод в файл <f_r>.res
*vwrite
(1X,' Нелинейный анализ')
*vwrite
(1X,' потери устойчивости пологой арки.')
```

```

*vwrite
(1X,' Nonlinear Buckling Analysis ')
*vwrite
(1X)
*vwrite,darc,hg,hl,bl
(1X,' DARC=', E12.5,' HG=', E12.5,' HL=', E12.5,' BL=',
E12.5)
*vwrite,rarc
(1X,' RARC=', E12.5)
*vwrite
(1X)
/output              ! Окончание сервисного вывода

/prep7
et,1,beam3          ! КЭ BEAM3
mp,ex,1,2.02e6      ! Модуль Юнга EX=2.02e6
mp,nuxy,1,0.3       ! Коэффициент Пуассона NUXY=0.3
r,1,bl*hl,(bl*hl**3)/12,hl ! R. const для поперечного сечения
балки:
                                ! площадь, момент инерции, толщина
csys,1              ! Переход в цилиндрическую систему
координат
k,1,rarc,90         ! Точка 1 - вершина купола
k,2,rarc,90-alfa    ! Точка 2 - основание арки
n,1,rarc,90         ! Узел 1 будет далее узлом
приложения силы,
n,2,rarc,90-alfa    ! узел 2 - основанием арки
l,1,2,nldiv         ! Генерация формы половины арки
! (В цилиндрической системе коорд. линии между точками
! с одинаковыми радиусами есть дуги окружности)
lmesh,all           ! Генерация сетки КЭ на линии

```

```

nummrg,node          ! Узлы 1,2 включаются в КЭ модель
finish
/solu
antype,static
nlgeom,on           ! Включение учета больших перемещений
                    ! (геометрической нелинейности)
arcl,on             ! Включение Arc-Length метода
nsub,50             ! Задание параметра NSUBST=50 для
                    ! определения радиуса в Arc-Length методе как
                    ! макс. значение нагрузки / NSUBSP
outres,all,all      ! Запись всех выходных данных
                    ! на всех шагах решения в базу данных
dk,1,ux,0,,,rotz    ! Условия симметрии в вершине купола арки
dk,2,ux,0,,,uy,rotz ! Жесткая заделка опорной точки 2 арки
!f,1,fy,-vforce     ! Задание сжимающей силы в верхней точке
(1) купола
sfbeam,all,,pres,vpres ! Задание давления
/psf,pres,norm,1     ! Показ распределенных нагрузок
eplot               ! Графический вывод конечно-элементной модели
solve               ! Запуск на счет
finish
/post1              ! Обработка результатов построителем post1
set,last            ! считывание результатов для последнего шага
pldisp,1            ! Изображение деформированного состояния,
                    ! наложенного на недеформированное

finish
/post26             ! Обработка результатов построителем post26
nsol,2,1,u,y        ! Определение переменной номер 2
как
                    ! вертикального перемещения uy в узле 1
abs,2,2             ! Переменная 2 := |переменная 2|
xvar,2              ! Выбор переменной номер 2 в качестве оси x
/axlab,x,Deflection ! Задание подписи Deflection под осью x
!prod,3,1,,,Force,,,2*vforce ! Умножение переменной 1 на
2*vforce
                    ! для получения значений силы
                    ! и присвоение переменной 1 метки Force
!/axlab,y,Force     ! Задание подписи Force под осью y
/axlab,y,Pressure   ! Задание подписи Force под осью y
prod,3,1,,,Pres,,,2*vpres ! Умножение переменной 1 на
2*vpres
                    ! для получения значений давления
                    ! и присвоение переменной 1 метки Pres
plvar,3              ! Вывод графика переменной 3 от 2
                    ! (кривой "нагрузка-прогиб")
/output,f_r,res,,append
*vwrite
(1X,' Значения точек кривой "нагрузка-прогиб"')
prvar,2,3            ! Печать значений точек предыдущего
графика
/output

```


Результатами выполнения программы Buckl_Nonlinear.inp являются график равновесных состояний «нагрузка-прогиб» (Рис. 3), где прогиб вычисляется в верхней точке купола, и значения точек этого графика в текстовом файле вывода <f_r>.res. По этим значениям находим, что величина критической нагрузки q , при которой происходит прохлопывание арки есть $q=133.557 \text{ кГ/см}^2$, что достаточно близко к величине критической нагрузки q_1 , полученной по программе Buckl_Eigenvalue.inp. (Такая близость результатов вовсе не характерна для задач устойчивости, в чем можно убедиться, проведя вычисления в программах Buckl_Eigenvalue.inp и Buckl_Nonlinear.inp для сосредоточенной силы.) Считывая данные для подшагов до прохлопывания арки и после прохлопывания, можно получить на ANSYS графики масштабированных деформируемых состояний до (рис. 2б) и после (рис. 2в) хлопка.




Рис. 2а.




Рис. 2б.




Рис. 2в.

Из сравнения рис. 2а и 2в видно, что форма волнообразования, представленная на рис. 2а после хлопка не реализуется.




Рис. 3.

Используя приведенные программы, аналогичным образом можно исследовать и другие задачи устойчивости деформируемых сред, в частности задачи о прохлопывании мембран и др. При этом необходимо выбрать подходящие КЭ ANSYS, убедившись, что они обладают требуемыми для

анализа устойчивости свойствами. Имея рабочие программы, можно также провести различные расчеты при варьировании входных параметров и убедиться в существенной зависимости получаемых результатов от параметров тонкостенности, пологости и др.